Newsgroups: vmsnet.networks.desktop.pathworks
Subject: RE: Remote PC/Mac Upgrades
From: Bob Findlay <FINDLAY@JII.AFRC.AC.UK>
Date: 25 Jan 1993 12:55:00 +0000 (GMT)
Content-transfer-encoding: 7BIT
Via: uk.ac.afrc.jii; Mon, 25 Jan 1993 12:56:02 +0000

sorry about the first message!
 
This is assuming you can't afford to buy any of the PW management products
currently on the market (One product I know of specifically, is LANutil from
Orion Europe, FAX num (UK) 021 311 1862 - which will probably do all you want)
 
We'd had PW for 2-3 years before hearing of this product, so have already
implemented all we wanted in BATch
 
We have a system that runs at startup time, and will run .BAT files if the
machine's node matches a preprepared list - we find we can do most updates this
way, detecting the machine's configuration with various home written or PD
utilities (is it 286?, 386?, EGA?, VGA?, EMS? etc)
 
As for configuring for new packages, all our supported programs are launched
via .BAT files, and we use the .BAT file to detect the setup for that program.
If it does not exist on that machine, then we do an auto install
 
I have included an example - Lotus Freelance
 
I hope this helps you.
 
Bob
P.S. I hope you're good at reading uncommented .BAT files!  As you can see, they
get pretty complicated - but that's the price you pay for doing everything on
the cheap!
 
FL.BAT
------
 
@ECHO OFF
ECHO FL.BAT 14DEC92
REM This is %sdrv%\FL.BAT  By P.Fretter NOV90 IPSR Computing Services
REM Tidy up  PF MAY92
REM Mod for FL4 BF 16OCT92
REM Mod to check for EMS PF 20OCT92
REM Mod to allow local default setup via environment variable NFL4=LDEFAULT
CALL c:\utils\virsrch.bat
IF EXIST c:\fl4setup\fl.com GOTO setOK
ECHO.
ECHO This PC has not yet been configured to use the network copy of Freelance
ECHO.
ECHO Automatic installation will now take place.
ECHO PLEASE WAIT.......
ECHO.
CALL %pdrv%\setupfl automatic
:setOK
ECHO.
TYPE %pdrv%\fl4\sbanner.txt
PRINTSER
ECHO.
ECHO Initialising - PLEASE WAIT........
ECHO.
USE %adrv% /d > nul
USE %adrv% \\%vaxnode%\fl4 /v
IF NOT ERRORLEVEL 1 GOTO Checkdef
ECHO No more connections to Freelance are available - please try again later
GOTO end
 
:checkdef
IF NOT "%NFL4%"=="LDEFAULT" GOTO CheckEMS
finddir \FL4SETUP\LDEFAULT > NUL
IF NOT ERRORLEVEL 1 GOTO defthere
ECHO Local default setup not present on this machine - unable to modify defaults
PAUSE
GOTO CheckEMS
 
:defthere
ECHO Initialising default setup - Please wait...
COPY \FL4SETUP\LDEFAULT \FL4SETUP > NUL
 
:CheckEMS
REM Check for existance of an EMS driver
chk4ems
IF ERRORLEVEL 1 GOTO NoEMS
 
:EMS
SET Small=393000
SET Norm=406000
SET Large=468000
REM LAT in EMS so no need to unload it
GOTO SetMemSize
 
:NoEMS
SET Small=453000
SET Norm=466000
SET Large=526000
REM Unload LAT to free as much RAM as possible
LAT /U >NUL
 
:SetMemSize
SET memsize=
REM Check if enough for Large mode
flmemchk %Large% >nul
IF ERRORLEVEL 1 GOTO normal
ECHO Loading Freelance in large drawing mode
SET memsize=l
GOTO run
:normal
flmemchk %Norm% > nul
IF NOT ERRORLEVEL 1 GOTO run
REM Check if enough for Small mode
flmemchk %Small% > nul
IF ERRORLEVEL 1 GOTO wawawawaah
 
 
ECHO Insufficient memory to run Freelance in full mode
ECHO Loading in reduced memory mode (13K less drawing space)
SET memsize=s
:run
attrib -r c:\fl4setup\*.*
SET oldpath=%path%
logdt FL4 IN %WSADDR% %NODE% [DD]/[MM]/[YY] [HH]:[MN]:[SS] ([DW]) >>
 P:\LOGS\FL.LOG
PATH=%sdrv%\;%adrv%\fl4
c:\fl4setup\fl %memsize%
SET memsize=
PATH=%oldpath%
attrib -r c:\fl4setup\*.*
logdt FL4 OUT %WSADDR% %NODE% [DD]/[MM]/[YY] [HH]:[MN]:[SS] ([DW]) >>
 P:\LOGS\FL.LOG
REM CLean up environment
SET oldpath=
USE %adrv%/d > nul
TYPE %pdrv%\fl4\fbanner.txt
IF EXIST %pdrv%\kwikkie.txt TYPE %pdrv%\kwikkie.txt
GOTO end
:wawawawaah
ECHO You have insufficient memory to run Freelance v4.0.  Exit all applications
ECHO and try again
:end
SET Small=
SET Norm=
SET Large=
REM load up LAT - doesn't matter if we've already done it
emsload LAT >NUL
ECHO.
 
SETUPFL.BAT
-----------
@ECHO OFF
IF %x%"==" GOTO Spawn
GOTO DoIt
 
:Spawn
SET x=Y
%comspec% /e:1024/c %pdrv%\setupfl.bat %1
SET x=
GOTO exit
 
:DoIt
REM If p1 exists then don't ask for user response...
IF NOT %1"==" GOTO GetDriver
 
ECHO        About to install JIC standard network setup of Freelance Plus v4.0N
ECHO.
ECHO                      The existing user setup will be deleted!!
ECHO.
ECHO.
ECHO                             To abort do CTRL-C else
ECHO.
PAUSE
 
:GetDriver
%pdrv%\getvdrv
IF ERRORLEVEL 10 GOTO BadDriver
IF ERRORLEVEL 9 GOTO VGAdetected
IF ERRORLEVEL 8 GOTO BadDriver
IF ERRORLEVEL 7 GOTO BadDriver
IF ERRORLEVEL 6 GOTO BadDriver
IF ERRORLEVEL 5 GOTO EGAdetected
IF ERRORLEVEL 4 GOTO EGAdetected
IF ERRORLEVEL 3 GOTO EGAdetected
IF ERRORLEVEL 2 GOTO BadDriver
IF ERRORLEVEL 1 GOTO BadDriver
GOTO BadDriver
 
:VGAdetected
SET Vdriver=VGA
GOTO Install
 
:EGAdetected
SET Vdriver=EGA
GOTO Install
 
:BadDriver
ECHO.
ECHO Your graphics adapter is neither EGA or VGA, so FL cannot be installed.
ECHO Please contact the Computing Department for advice.
ECHO.
GOTO Exit
 
:Install
 
ECHO Installing for a %Vdriver% graphics driver
IF NOT EXIST c:\fl4setup\*.* GOTO NotInst
GOTO RemoveIt
:NotInst
MD c:\fl4setup
GOTO CopyIt
 
:RemoveIt
ECHO Removing existing setup.....
attrib -r c:\fl4setup\*.* >NUL
DEL c:\fl4setup\*.* < %pdrv%\stdfl4\y.txt >NUL
 
:CopyIt
ECHO Installing Freelance 4.0N.....
COPY %pdrv%\stdfl4\%Vdriver%\*.* c:\fl4setup >NUL
attrib +r c:\fl4setup\*.* >NUL
 
ECHO FL 4.0N successfully installed.
ECHO.
 
:Exit